EE 230 Experiment - 8

Special Opamp Linear Circuits - Precision Rectifiers and Active Filters

1st October, 2021

Vinamra Baghel 190010070

1 Precision Rectifiers

1.2 Improved Half Wave Rectifier - A

```
Vinamra Baghel 190010070 Improved Half Wave Rectifier-A
.include ua741.txt
.include IN914.txt
*Netlist
r1 in neg 10k
r2 neg out 10k
rl out gnd 1k
D1 neg o1 1N914
D2 o1 out 1N914
X1 gnd neg pp np o1 ua741
Vccp pp gnd 12
Vccn np gnd -12
Vin in gnd sin(0 1 100 0 0 0)
*Analysis
.tran 1u 40m
.control
run
plot v(out) vs v(in)
.endc
.end
```



Simulation



(a) V_{in} and V_{out}



(b) V_{out} vs V_{in}

Learnings:

Learnt about the Improved Half-wave rectifier of type A. This rectifier allows the negative part of the input wave pass and outputs it as a positive wave. The model contains diodes D_1 and D_2 which help current pass through the OpAmp in both positive and negative cycles and prevent saturation and delay.

During the positive half-cycle of V_i, D_1 is on and D_2 is off; the current flows into the OpAmp; V_o is 0. During the negative half-cycle of V_i, D_2 is on and D_1 is off; the current flows from the OpAmp; V_o is $-V_i$.

We also saw the OpAmp output terminal, V_{o1} behaves during operation. We can see an offset voltage which is equal to the voltage drop across the diode.

1.4 Full Wave Rectifier

```
Vinamra Baghel 190010070 Full Wave Rectifier
.include ua741.txt
.include IN914.txt
*Netlist
r1 in neg1 10k
r2 neg1 in1 10k
rs1 in neg2 10k
rs2 in1 neg2 5k
rf neg2 out 10k
rl out gnd 1k
D1 o1 neg1 1N914
D2 in1 o1 1N914
X1 gnd neg1 pp np o1 ua741
X2 gnd neg2 pp np out ua741
Vccp pp gnd 12
Vccn np gnd -12
Vin in gnd sin(0 1 100 0 0 0)
*Analysis
.tran 1u 40m
.control
run
plot v(out) vs v(in)
.endc
.end
```



Simulation



V_{in} and V_{out}



V_{in} vs V_{out}

Learnings:

Learnt about the Full-wave Rectifier. The full-wave rectifier consists of a Improved half-wave rectifier-B and an inverting summer in cascade. One input of the summer is V_i and the other is the half-wave rectified wave.

2 Active Filters

2.2 Single-pole Active High-pass Filter

```
Vinamra Baghel 190010070 Single-pole Active High-pass Filter
.include ua741.txt
*Netlist
r1 out neg 9.1k
r2 neg gnd 1k
ra pos gnd 4.7k
ca in pos 0.1u
X1 pos neg pp np out ua741
Vccp pp gnd 12
Vccn np gnd -12
Vin in gnd dc 0 ac 1
*Analysis
.ac dec 100 1 10k
.control
run
plot vdb(out)
.endc
.end
```



Single-pole Active High-pass Filter

Simulation

The cut-off frequency is given by $f_c = \frac{1}{2\pi R_A C_A} = 338.63\,Hz$. The slope of the curve is +20 dB. We can see that both these value check out from the graph.



V_{out} in dB

Learnings:

Learnt about the Single-pole Active High-pass Filter. It has many advantages over a passive high-pass filter in terms of flexibility and gain. It is essentially a passive filter cascaded with an OpAmp.

2.4 Sallen-Key (2-pole) Active Low-pass Filter

```
Vinamra Baghel 190010070 Sallen-Key (2-pole) Active Low-pass Filter
.include ua741.txt
*Netlist
r1 out neg 1.8k
r2 neg gnd 3.3k
ra in 1 4.7k
rb 1 pos 4.7k
ca 1 out 0.1u
cb pos gnd 0.1u
X1 pos neg pp np out ua741
Vccp pp gnd 12
Vccn np gnd -12
Vin in gnd dc 0 ac 1
*Analysis
.ac dec 100 1 10k
.control
run
plot vdb(out)
.endc
.end
```



Sallen-Key (2-pole) Active Low-pass Filter

Simulation

The cut-off frequency is given by $f_c = \frac{1}{2\pi\sqrt{R_A R_B C_A C_B}} = 338.63\,Hz$. The slope of the curve is -40 dB. We can see that both these value check out from the graph.



V_{out} in dB

Learnings:

Learnt about the Sallen-key (2-pole) Active Low-pass Filter. It is essentially a 2-pole passive filter cascaded with an OpAmp. The transfer function obtained is of the order 2.

Files associated: https://github.com/VNMR-35/EE-230-Lab