EE 230 Experiment - 4

Current Source, Current Mirror, and Differential Pair

20th August, 2021

Vinamra Baghel 190010070

1 BJT Current Source

```
Vinamra Baghel 190010070 BJT Current Source
.model bc557a PNP IS=10f BF=100 ISE=10.3f IKF=50m NE=1.3 BR=9.5 VAF=80 IKR=12m
   ISC=47p NC=2 VAR=10 RB=280 RE=1 RC=40 tr=0.3u tf=0.5n cje=12p vje=0.48 mje
   =0.5 cjc=6p vjc=0.7 mjc=0.33 kf=2f
.SUBCKT DI_1N4734A 1 2
D1 1 2 DF
DZ 3 1 DR
VZ 2 3 4.48
.MODEL DF D ( IS=73.6p RS=0.620 N=1.10 CJO=165p VJ=0.750 M=0.330 TT=50.1n )
.MODEL DR D ( IS=14.7f RS=0.256 N=1.49 )
.ENDS
*Netlist
re cc e 4.7k
rb b gnd 2.2k
rl out gnd 1k
X b cc DI_1N4734A
Q c b e bc557a
Vc c out 0
Vcc cc gnd 12
*Analysis
.dc RL 1k 10k 1k
.control
run
plot i(Vc) vs V(out)
.endc
.end
```

Analysis

$I_E = (12 - (12 - 5.6 + 0.7))/4700 = I_L = 1.04$ mA. $V_L = I_L * R_L = 0.104$ V.
For R_{L-max}, $V_L = V_{CEsat} - V_E = 6.9$ V. Therefore, $R_{L-max} = 6.63\,k\Omega$.

Simulation

Operating point for $R_L = 1k\Omega$, $V_L = 1.722$ V



Figure 1: I_L vs V_L

Learnings:

V_L increases with R_L, hence for a maximum value of R_L, the device saturates.

2 BJT Current Mirror based Current Source

```
Vinamra Baghel 190010070 BJT Current Mirror based Current Source
.MODEL CA3046 NPN (IS=10.000E-15 BF=100 VAF=80 IKF=46.747E-3 ISE=114.23E-15 NE=1.4830 BR=.1001 VAR=100 IKR=10.010E-3 ISC=10.000E-15 RC=10 CJE=1.0260E-12 MJE=.33333 CJC=991.79E-15 MJC=.33333 TF=277.09E-12 XTF=309.38 VTF=16.364 ITF=1.7597 TR=10.000E-9)
*Netlist
r cc b 10k
Q1 b b gnd CA3046
Q2 out b gnd CA3046
Vo out gnd 1
Vcc cc gnd 12
*Analysis
.dc Vo 1 5 0.5
.control
run
print -i(Vo)
plot -i(Vo)
.endc
.end
```

Analysis

$$I_{ref} = \frac{V_{cc} - V_{BE}}{R} = \frac{12 - 0.7}{10000} = 1.13\,mA$$

$$I_o = I_{ref} . \frac{1}{1 + \frac{2}{\beta}} . \left(1 + \frac{V_o - V_{BE}}{V_A}\right)$$

1. $V_A = 80V$. For $V_o = 1$ V, $I_o = 1.112$ mA. For $V_o = 5$ V, $I_o = 1.167$ mA.
2. $V_A = 200V$. For $V_o = 1$ V, $I_o = 1.109$ mA. For $V_o = 5$ V, $I_o = 1.132$ mA.

Simulation

For $V_o = 1$ V, $I_o = 1.11$ mA.



(a) $V_A = 80$ V



(b) $V_A = 100$ V

Learnings:

I_o increases linearly with V_o. The graph's slope and intercept both decrease with increase in V_A.

3 Differential Pair

3.1 DC Operating Point

```
Vinamra Baghel 190010070 Differential Pair
.model bc547a NPN IS=10f BF=200 ISE=10.3f IKF=50m NE=1.3 BR=9.5 VAF=80 IKR=12m
    ISC=47p NC=2 VAR=10 RB=280 RE=1 RC=40 tr=0.3u tf=0.5n cje=12p vje=0.48 mje
    =0.5 cjc=6p vjc=0.7 mjc=0.33 kf=2f
*Netlist
re ee e 10k
rb1 in1 b1 1k
rb2 in2 b2 1k
rc1 c1 cd1 6.8k
rc2 c2 cd2 6.8k
Q1 c1 b1 e bc547a
Q2 c2 b2 e bc547a
Vcd1 cd1 cc 0
Vcd2 cd2 cc 0
Vin1 in1 gnd 0
Vin2 in2 gnd 0
Vee gnd ee 12
Vcc cc gnd 12
*Analysis
.op
.control
run
print v(e) v(c1) v(c2) (-i(Vee)) (-i(Vcd1)) (-i(Vcd2))
.endc
.end
```

Analysis

$V_E = 0 + V_{in} - V_{BE} = -0.7$ V (Approximating base currents to be 0). $V_{C1} = V_{C2} = V_{CC} - I_{C1} * R_{C1} = 8.158$ V.

$I_E = \frac{V_E - V_{EE}}{R_E} = \frac{-0.7-(-12)}{10000} = 1.13$ mA. $I_{C1} = I_{C2} = I_E/2 = 0.565$ mA (Approximating base currents to be 0).

Simulation

$V_E = -0.646$ V
$V_{C1} = V_{C2} = 8.171$ V
$I_E = 1.135$ mV
$I_{C1} = I_{C2} = 0.563$ mA

3.2 Differential Amplifier - Single Stage

```
Vin1 in1 gnd sin(0 10m 1k 0 0)
.tran 0.01m 2m
.control
run
plot v(in1)*100 v(c1) v(c2)
meas tran a pp v(c1)
meas tran b pp v(in1)
print a/b
```

Analysis

$g_m = I_C/V_T = 0.0217\,\Omega^{-1}$.
Gain $= \frac{-g_m.R_C}{2} = \frac{-0.0217*6800}{2} = 73.78$.

Simulation

Gain $= \frac{V_{C1-peak-to-peak}}{V_{in1-peak-to-peak}} = \mathbf{56.68394}$.



Figure 3: $V_{in1} * 100$, V_{C1}, V_{C2}

3.3 Large-Signal Characteristics (Voltage-Transfer Characteristics - VTC)

```
Vin1 in1 gnd -1
.dc Vin1 -1 1 0.01
.control
run
plot v(c1) v(c2) vs v(in1)
```

Simulation



Figure 4: V_{C1} and V_{C2} vs V_{in1}

Learnings:

We get amplified sinusoidal waveforms at both O1 and O2 that differ by a phase of π. These waveforms are centred at the operating point voltage calculated in Section 3.1, which is **8.171 V**.

Files associated: https://github.com/VNMR-35/EE-230-Lab